Week 9

Ch 13: Monads

University of the Fraser Valley

Dr. Russell Campbell

Russell.Campbell@ufv.ca

COMP 481: Functional and Logic Programming

Overview

- Intro to Monads
- Tightrope Walking Simulation (Pierre)
- Banana on a Wire
- `do` Notation
- Pierre Returns
- Pattern Matching and Failure
- The List Monad
- `MonadPlus` and the `guard` Function
- A Knight's Quest
- Monad Laws
  - Left Identity
  - Right Identity
  - Associativity
- More Simplifications

Intro to Monads

Bind `>>=`

We have worked with implementing applicatives for various types so that:

- `Maybe a` values represent computations that may end up in failure
- `[a]` values represent all possible computational results (nondeterministic)

- `IO a` values represent computations with side effects

These can be facilitated with the special characters `>>=` as a binary operation between Monad values. This function is called bind.

Monad

Monads are a type class with similar behaviour as `Functors` and `Applicatives` to make functions work in context:

```
(>>=) :: (Monad m) => m a -> (a -> m b) -> m b
```


This time, we want:

- to take an input value with some context `m a`
- a function that expects no input context `a ->`
- but the function returns a result `m b` with context when applied on the input `m a`

Context of `Maybe`

Recall how we mapped with functors:


```
ghci> fmap (++"!") (Just "wisdom")
Just "wisdom!"
```



```
ghci> fmap (++"!") Nothing
Nothing
```


- a value of `Nothing` as a result of such a mapping can be interpreted as a failure for some calculation

Context with Applicative

Applicative functors have the added context to the function as well:

```
ghci> Just (+3) <*> Just 3
Just 6
```



```
ghci> Nothing <*> Just "greed"
Nothing
```



```
ghci> Just (ord) <*> Nothing
Nothing
```

- if either of the operands is `Nothing` it is propagated to the result

Applicative <$> and <*>

There was also the applicative style:


```
ghci> max <$> Just 3 <*> Just 6
Just 6
```



```
ghci> max <$> Just 3 <*> Nothing
Nothing
```

Monad Implementation

Now the implementation of the `Monad` type class:

```
class Applicative m => Monad m where
    return :: a -> m a

    (>>=) :: m a -> (a -> m b) -> m b

    (>>) :: m a -> m b -> m b
    x >> y = x >>= \_ -> y
```

return

- the `return` function is the same as `pure` as we saw it in the `Applicative` type class
  - recently Haskell developers decided it would be a requirement to make any `Monad` also be a subclass of `Applicative`
  - just a reminder that `return` is not like in other programming languages—here it simply wraps a value within the context of `m`
- the `>>` operation has a default implementation that is rarely changed
- there also used to be a `fail` function, but that is no longer required to implement an instance of `Monad`

Maybe as a Monad

The `Maybe` type as an instance of `Monad`:

```
instance Monad Maybe where
    return x = Just x
    Nothing >>= f = Nothing
    Just x >>= f = f x
```


- if there is `Nothing` input on the left-hand side of `>>=`, the expression evaluates also to `Nothing`
- otherwise, there is a nested value within `Just` and we can apply the function `f` to it
  - note that the result of `f` is in a context with a nested value that at least has the same type as `x`

Example Maybe Monad

Now we give `Maybe` a try as a monad:

```
ghci> return "WHAT" :: Maybe String
Just "WHAT"

ghci> Just 9 >>= \x -> return (x*10)
Just 90

ghci> Nothing >>= \x -> return (x*10)
Nothing
```

— Tightrope Walking Simulation —

Using Monads

We will demonstrate one of the advantages of monads:

- change the behaviour of calculations in the way we desire

- context of a monad can participate in the computation

- we cannot do this with applicatives alone, since they only lift computations into the nested context

Tightrope Walking

Suppose we have a man Pierre that tightrope walks with a pole:

- birds land on either side of the pole

- if more than a difference of 3 birds land on either side, Pierre falls…

(to a safety net, of course)



Photographer: Grant Gibson (CC BY 3.0)

Simulation of Birds

We will first implement a few types to help us keep track of the number of birds:

```
type Birds = Int

type Pole = (Birds, Birds)
```


Next we want functions to simulate birds landing on either side of the pole:

```
landLeft :: Birds -> Pole -> Pole
landLeft n (left, right) = (left + n, right)


landRight :: Birds -> Pole -> Pole
landRight n (left, right) = (left, right + n)
```

Without Monads

We try out our functions without monads:

```
ghci> landLeft 2 (0, 0)
(2, 0)
ghci> landRight 1 (1, 2)
(1,3)
ghci> landRight (-1) (1, 2)
(1,1)
```


- just use a negative number to simulate birds flying away

Order of Operations

Chain simulated birds landing by nesting operations:

```
ghci> landLeft 2 (landRight 1 (landLeft 1 (0, 0)))
(3,1)
```


- we can create a utility function to help us write more concisely:


```
x -: f = f x
```


- then we can write the parameter before the function, and rewrite our previous expression:

```
ghci> (0, 0) -: landLeft 1 -: landRight 1 -: landLeft 2
(3,1)
```

Maybe to Manage Failure (1)

So far, this does not check our condition for if Pierre will fall

- if we did any lopsided landing of birds, then we just get back the ordered pair
- instead, we would like to check for Pierre's failure, so we use `Maybe`

Maybe to Manage Failure (2)

Implement new versions of our bird-landing simulation functions:

```
landLeft :: Birds -> Pole -> Maybe Pole
landLeft n (left, right)
 | abs (left + n - right) < 4   = Just (left + n, right)
 | True                         = Nothing

landRight :: Birds -> Pole -> Maybe Pole
landRight n (left, right)
 | abs (left - right - n) < 4   = Just (left, right + n)
 | True                         = Nothing
```

Simulating Imbalance

- Our implementation will maintain a difference of three for the number of birds on either side of the pole
- if > 3, the result of any `landLeft` or `landRight` will be `Nothing` to indicate imbalance and represent falling
- since these versions of our functions:
  - take a `Pole` for input,
  - but a `Maybe Pole` for output,
  - we will need to make use of `>>=`
  - to apply successive operations together


```
ghci> return (0, 0) >>= landLeft 1 >>= landRight 1 >>= landLeft 2
Just (3,1)
```

Using >>=

```
return (0, 0) >>= landLeft 1 >>= landRight 1 >>= landLeft 2
```

- note that we had to begin the calculation with the context of a monad, so we used `return`
- the `return` function can be used no matter the specific application of a monad context for a sequence of calculations

Finally, let's see Pierre fall!


```
ghci> return (0, 0) >>= landLeft 1 >>= landRight 4
        >>= landLeft (-1) >>= landRight (-2)
Nothing
```

- can you tell at which point the pole became imbalanced?

Know the Monad

Try to make sure you do not conflate the context of the monad with the functions used!

- monad is `Maybe` and NOT the functions `landLeft` and `landRight`
- the functions have merely been edited to take advantage of `Maybe` as a monad

— Banana on a Wire —

Banana Slip

We implement more functions that can combine with the other computations we have designed for simulation:

- suppose a banana on the wire could slip Pierre while walking
- this automatically forces Pierre to fall


```
banana :: Pole -> Maybe Pole
banana _ = Nothing
```


It is fairly clear what will happen when we use this function:

```
ghci> return (0, 0) >>= landLeft 1 >>= banana >>= landRight 1
Nothing
```

Changing Default >>

To ignore a monadic value on the *right* and *return the left* value, we can adjust the `>>` operation from its default:

```
(>>) :: (Monad m) => m a -> m b -> m a
n >> m = m >>= \_ -> n
```


Otherwise, the default is to ignore the *left* value and *return the right* value, equivalent to the `do` block:


```
do
  n
  m
```

for monad values `n` and `m`, the above returns `m`.

Carrying Monads Forward

```
ghci> Nothing >> Just 3
Nothing

ghci> Just 3 >> Just 4
Just 4

ghci> Just 3 >> Nothing
Nothing
```


(keep in mind the above demonstrates the default implementation!)

Thus, we can omit having to write a `banana` function,
and just use `>> Nothing` to the same effect.

`do` Notation

>>= with Lambdas

We can use monad-style expressions with lambdas:

```
ghci> Just 3 >>= (\x -> Just (show x ++ "!"))
Just "3!"
```

- monadic value `Just 3` has its nested `3` passed as input into the lambda on the right side
- a monadic value is returned `Just "3!"`

The above expression can be rewritten as two nested `>>=` operations:

```
ghci> Just 3 >>= (\x -> Just "!" >>= (\y -> Just (show x ++ y)))
Just "3!"
```

Notice >>= “binds” an unwrapped value to the parameter.

Binding and Nesting

The expression can be rewritten as two nested `>>=`:

```
ghci> Just 3 >>= (\x -> Just "!" >>= (\y -> Just (show x ++ y)))
Just "3!"
```


Notice `>>=` “binds” an unwrapped value to the parameter.

- this is similar to the following:


```
let
    x = 3;
    y = "!"
in show x ++ y
```

Helpful But Less Readable

The advantage of the more elaborate version:

- we get monads to help manage context
- at each part of the calculation
- without needing to explicitly write code at each stage to deal with it


```
ghci> Nothing >>= (\x -> Just "!" >>= (\y -> Just (show x ++ y)))
Nothing
```



```
ghci> Just 3 >>= (\x -> Nothing >>= (\y -> Just (show x ++ y)))
Nothing
```



```
ghci> Just 3 >>= (\x -> Just "!" >>= (\y -> Just Nothing))
Just Nothing
```

- at each point, the value could instead be `Nothing`,
  and the result is dealt with appropriately without error

Organized as a Function

We move toward a nicer syntax available, first, in the form of a function:


```
:set +m

let
foo :: Maybe String
foo = Just 3
    >>= (\x -> Just "!"
    >>= (\y -> Just (show x ++ y)
    ))
```

Maybe Context (1)

- there is an alternative cleaner syntax available with the `do` block


```
foo :: Maybe String
foo = do
    x <- Just 3
    y <- Just "!"
    Just (show x ++ y)
```

Maybe Context (2)

```
foo :: Maybe String
foo = do
    x <- Just 3
    y <- Just "!"
    Just (show x ++ y)
```

- the `do` block allows a different way to chain monadic calculations into one monadic calculation
- if any of the monadic values is a `Nothing` then the result of the `do` expression will be `Nothing`
- lines that are not monadic value have to be in a `let` expression
- we use `<-` assignment to obtain a nested value (bind)
  - if we have a `Just "!"` monadic value, the nested value is `"!"` as a `String` type
  - if we have a `Just 3` monadic value, the nested value is `3` as a numeric type
- the last line of a `do` block cannot use `<-`, since this would not make sense as the result returned for a monadic expression

Typical Do Block

The typical design:

- to compute and assign nested values
- on each line of the `do` block

- and return them in some combined expression

- within the monadic context

Equivalent Examples

One more small example:

```
ghci> Just 9 >>= (\x -> Just (x > 8))
Just True
```



```
let
  marySue :: Maybe Bool
  marySue = do
    x <- Just 9
    Just (x > 8)

ghci> marySue
Just True
```

Review

(Simranjit Singh)

```
-- various types of addition

-- infix (any func that's a special symbol is automatically infix)
1 + 2

-- prefix
(+) 1 2

-- functor
fmap (+1) [1,2,3]
(+1) <$> [1,2,3]

-- applicative functor
[(+1)] <*> [1,2,3]
[(+)] <*> [1] <*> [1,2,3]
pure (+) <*> [1] <*> [1,2,3]
(+) <$> [1] <*> [1,2,3]
```

Examples

(Simranjit Singh)

```
-- monads
[1] >>= \x -> return (x+1)
[1,2,3] >>= \x -> return (x+1)

-- nested >>= operations
[1,2,3] >>= \x -> return (x+1) >>= \y -> return (y+1)

-- alternative do block
do
    x <- [1,2]
    y <- [3,4,5]
    return $ x + y + 1
```

Pierre Returns

Simulation with Do Block

We can rewrite our previous example of Pierre's tightrope walking with a simulation for birds landing on a pole.

We now design it in a `do` block:

```
routine :: Maybe Pole
routine = do
    start <- return (0, 0)
    first <- landLeft 2 start
    second <- landRight 2 first
    landLeft 1 second
```



```
ghci> routine
Just (3,2)
```

- each line of a `do` block depends on the success of the previous one

Nested Cases

Without monads, this issue can be seen differently where computation would have to be *nested*:

```
routine :: Maybe Pole
routine = case Just (0, 0) of
    Nothing -> Nothing
    Just start -> case landLeft 2 start of
        Nothing -> Nothing
        Just first -> case landRight 2 first of
            Nothing -> Nothing
            Just second -> landLeft 1 second
```


- the ghci session will issue a warning with the above code, but you should still be able to issue `routine`

Nothing Overwrites Results

Then if we want to throw in a banana peel like we did before:

```
routine :: Maybe Pole
routine = do
    start <- return (0, 0)
    first <- landLeft 2 start
    Nothing
    second <- landRight 2 first
    landLeft 1 second
```


- the line with `Nothing` does not use `<-`, much like our use of `>>` to ignore a previous monadic value
    - this is nicer than needing to write equivalently `_ <- Nothing`

>>= vs Do

It is up to you whether you want to use `>>=` versus `do` blocks, but in general:

- to avoid naming prior results, use `>>=`
- to mix together multiple previous results, use `do` blocks

Neither is exclusively needed to accomplish the above…

— Pattern Matching and Failure —

Bind with Pattern Matching

Pattern matching can be used on a binding:

```
justH :: Maybe Char
justH = do
    (x:xs) <- Just "hello"
    return x
```


- the above grabs the first letter of the string `"hello"`
- the `` `justH` `` function evaluates to `` `Just h` ``
  - remember, the left value of a `` `:` `` operation is a `` `Char` ``, not a singleton

Failing a Pattern Match

When a pattern match fails within a function:

- the next pattern is attempted
- if matching falls past all patterns, the function throws an error
  - the program crashes

With `let` expressions, an error occurs on failure of matching because there is no falling mechanism for matching further patterns.

Implementing `fail` Function

When matches fail within a `do` block:

- the context of the monad often implements a `fail` function
- to deal with the issue in its context
- as we have seen with the `Maybe` type

- this used to be implemented as part of a default `Monad` function
- it is now dealt with as an instance of the `Monad` type with a custom implementation of `fail` per each type

For example, with `Maybe` as a `Monad`, we can implement:

```
fail :: Maybe a -> Maybe a
fail _ = Nothing
```

- but `fail` is a default function to throw an error with String message
- then when all patterns fall through unmatched within a `do` block, the function expression will evaluate to `Nothing` *instead of crashing*

Example of `fail`

```
wopwop :: Maybe Char
wopwop = do
    (x:xs) <- Just ""
    return x
```



```
ghci> wopwop
Nothing
```

- there is only a failure mitigated within the context of monad `Maybe`
- there is no program-wide failure

-- The List Monad --

Lists as Monads

Recall that we can do nondeterministic calculations with lists using the applicative style:

```
ghci> (*) <$> [1,2,3] <*> [10, 100, 1000]
[10,100,1000,20,200,2000,30,300,3000]
```


Let us now see the implementation of `Monad` for lists:


```
instance Monad [] where
    return x = [x]
    xs >>= f = concat (map f xs)
    fail _ = []
```


- `return` just puts the input value within minimal list contex, i.e.: a singleton `[x]`

List Context

The function `concat` might seem not to fit the context, but we want to implement nondeterminism.

```
ghci> [3,4,5] >>= \x -> [x,-x]
[3,-3,4,-4,5,-5]
```


- as you can see, all the possible results of `[3,4,5]` fed into `\x -> [x, -x]` are shown as one conjoined list

The `>>=` operation can handle `[]`:

```
ghci> [] >>= \x -> ["bad", "sad"]
[]
ghci> [1,2,3] >>= \x -> []
[]
```

Chaining >>=

It is possible to chain `>>=` operations to propagate the nondeterminism:

```
ghci> [1,2] >>= \n -> ['a','b'] >>= \ch -> return (n, ch)
[(1,'a'),(1,'b'),(2,'a'),(2,'b')]
```


- notice that the input variable `n` shows up as part of the final expression after the next `>>=` operation
  - remember, each next `>>=` operation is nested as part of the previous one
- `return` places each pair within a singleton context
- all the pairs are concatenated together into one flat list

Using Chaining

Describing the propagation of nondeterministic operations:

- "for all" elements in `[1,2]` should be paired
- with every element of `['a','b']`.

The previous expression could be written in a `do` block, but you might find the following syntax easier to read:


```
:{
[1,2]
    >>= \n -> ['a','b']
    >>= \ch -> return (n, ch)
:}
```



```
[(1,'a'),(1,'b'),(2,'a'),(2,'b')]
```

Chaining in a `do` Block

Otherwise, in a module, I would use a `do` block:

```
listOfTuples :: [(Int, Char)]
listOfTuples = do
    n <- [1,2]
    ch <- ['a','b']
    return (n, ch)
```



```
ghci> listOfTuples
[(1,'a'),(1,'b'),(2,'a'),(2,'b')]
```

- these syntax make the nondeterminism clearer to keep track of
  - `n` takes on every value of `[1,2]`
  - `ch` takes on every value of `['a','b']`

Similar to List Comprehension

Lastly, we had originally learned list comprehension to do essentially the same thing as above:

```
ghci> [ (n, ch) | n <- [1,2], ch <- ['a','b'] ]
[(1,'a'),(1,'b'),(2,'a'),(2,'b')]
```


- the `<-` notation works pretty much the same, to handle the nondeterministic context for us
- we did not need to use the `return` function because list comprehension takes care of that for us
- documentation typically calls alternatives such as this syntactic sugar for the more formally written expressions

— `MonadPlus` and the `guard` Function —

Monad Filtering

List comprehension can apply filtering with a conditional expression:

```
ghci> [ x | x <- [1..50], '7' `elem` show x ]
[7,17,27,37,47]
```


The `MonadPlus` type class is for implementing filtering.

- it is for monads that can also act as monoids


```
class Monad m => MonadPlus m where
    mzero :: m a
    mplus :: m a -> m a -> m a
```


- `mzero` is synonymous with `mempty` from `Monoid`
- `mplus` corresponds to `mappend`

MonadPlus

We know lists are both monads as well as monoids, so:


```
instance MonadPlus [] where
    mzero = []
    mplus = (++)
```


- a failed computation for lists is an empty list
- `mplus` concatenates two nondeterministic computational results

Filtering (1)

There is also a `guard` function that helps perform filters:

```
import Control.Monad


guard :: (MonadPlus m) => Bool -> m ()
guard True = pure ()
guard False = mzero
```


- a Boolean expression is input into `guard` as the test to either create a dummy value or nothing (`mzero`)
- empty tuple `pure ()` is used as a dummy and used to then filter
  - input into `>>` operations on the left side, it will either keep or throw away the right-hand side values

Filtering (2)

```
ghci> import Control.Monad

ghci> guard (5 > 2) >> return "cool" :: [String]
["cool"]

ghci> guard (1 > 2) >> return "cool" :: [String]
[]
```

Using guard

There are two ways we can write the use of `guard` in order to filter as in the list comprehension:

- the first is with nested `>>=` expressions
- the second is within a `do` block

```
ghci> [1..50] >>= (\x -> guard ('7' `elem` show x) >> return x)
[7,17,27,37,47]
```



```
let
sevensOnly :: [Int]
sevensOnly = do
    x <- [1..50]
    guard ('7' `elem` show x)
    return x

ghci> sevensOnly
[7,17,27,37,47]
```

Examples

(David Semke)

```haskell
import Control.Monad

-- Using list1, create all possible pairs (x, y)
--   such that x is always greater than y

list1 = [1, 2, 3, 4, 5]

listCompPairs =
  [(x, y) | x <- list1, y <- list1, x > y]

nestedMethodPairs =
  list1 >>= (\x -> list1 >>= (\y -> guard (x > y) >> return (x, y)))

doMethodPairs = do
    x <- list1
    y <- list1
    guard (x > y)
    return (x, y)
```

— A Knights Quest —

Simulating Knights in Chess

We would like to simulate on a chess board, a knight which has a restricted `L` move each turn.

Are they able to reach a square within three turns?

- the image below shows the positions in one turn where a knight piece could choose to move
  - it should be symmetrical, and there are two spots missing behind the first row, but the pieces cannot move off the board



Pairs for Positions

We use a pair to keep track of the row and the column

- the first number gives the row
- the second number gives the column


```
type KnightPos = (Int, Int)
```


So, if the knight starts at position `(6, 2)`, can they move to `(6, 1)`?

- we might wonder which is the best move to choose toward the goal
- instead, we just let nondeterminism try all of the moves

moveKnight

```
moveKnight :: KnightPos -> [KnightPos]
moveKnight (c, r) = do
    (c', r') <- [
            (c+2, r-1), (c+2, r+1), (c-2, r-1), (c-2, r+1),
            (c+1, r-2), (c+1, r+2), (c-1, r-2), (c-1, r+2)
        ]
    guard (c' `elem` [1..8] && r' `elem` [1..8])
    return (c', r')
```



```
ghci> moveKnight (6, 2)
[(8,1),(8,3),(4,1),(4,3),(7,4),(5,4)]
ghci> moveKnight (8, 1)
[(6,2),(7,3)]
```

- we can filter the new positions with use of `guard`

`in3` Possibilities

Next, we can use this to write a concise function to move three times:

```
in3 :: KnightPos -> [KnightPos]
in3 start = do
    first <- moveKnight start
    second <- moveKnight first
    moveKnight second
```


Passing in `(6, 2)` generates a fairly long list:

```
ghci> in3 (6, 2)
```

(results omitted for space)

Using `in3` Output

We could rewrite the `in3` function using `>>=` notation:

```
in3 start =
 return start >>= moveKnight >>= moveKnight >>= moveKnight
```


- the `return` puts `start` within the context of lists (or nondeterminism)

Finally we can test for whether `(6, 2)` is an element in the result:


```
ghci> (6, 2) `elem` in3 (6, 1)
True
ghci> (6, 2) `elem` in3 (7, 3)
False
```

Extending Chess Simulation

We could write the previous movement testing as a function and pass in the start and end positions.

- (the next chapter shows how to modify the above as a function that can also give back the possible moves to take)

- we could also specify how many moves in general as input, not just three moves, specifically

— Monad Laws —

Monad Laws

Each rule expects two equivalent expressions:

— Left Identity —

- `return x >>= f`
- `f x`

— Right Identity —

- `m >>= return`
- `m`

— Associativity —

- `(m >>= f) >>= g`
- `m >>= (\x -> f x >>= g)`

Left Identity

- `return x >>= f`
- `f x`

Remember, that in the situation of monads, the function `f` will result in a value with context.

- note that `return` wraps with that context, and `>>=` removes context to pass the nested value to `f`


```
f :: Num a => a -> Maybe a
f x = Just (x+100000)

ghci> return 3 >>= f
Just 100003

ghci> f 3
Just 100003
```

Right Identity

- `m >>= return`
- `m`

Consider right side of first expression:

- function `return` takes a value and wraps it in a minimal context
- for `Maybe` type, minimal context "does not introduce any failure"
- for lists, minimal context "does not introduce extra nondeterminism"

With lists, say if we feed `[1,2,3]` into `return` with `>>=`:

- first, every element of the list gets wrapped, to get `[[1],[2],[3]]`
- the elements concatenate with `(++)` applied to result in `[1,2,3]`

Associativity

- `(m >>= f) >>= g`
- `m >>= (\x -> f x >>= g)`

The order that operations executed in a sequence should not matter.

The functions `f` and `g` could be composed first

- but the notation for lambda expression is the resulting composed function,
- instead of something a bit more concise as in mathematics as with $(g \circ f)$
  (yes, the order is correct with the above monad law)
- but notice Haskell syntax makes sense for order of execution when we are writing our code

Chaining and Associativity

Recall that we had simulated tightrope walking, and chained `>>=` expressions as with the law of associativity:

```
pure (0, 0) >>= landRight 2 >>= landLeft 2 >>= landRight 2
Just (2,4)
```


- the use of `pure` we have used with `>>=` before
- it reads a bit nicer than using `return` at the start of a block of code

The law of associativity allows us to drop parentheses,
but with parentheses, we have:

```
((pure (0, 0) >>= landRight 2) >>= landLeft 2) >>= LandRight 2
Just (2,4)
```

Multiline

But we can also write the expression as:

```
:{
pure (0, 0)
  >>= (\x -> landRight 2 x
  >>= (\y -> landLeft 2 y
  >>= (\z -> landRight 2 z
  )))
:}
Just (2,4)
```


- each successive function is further nested in parentheses

Flipping with `<=<`

At least the law of associativity allows us to be very concise and avoid excessive use of parentheses.

The following operation flips use of `>>=` for nesting functions that work with monads together.

```
(<=<) :: (Monad m) => (b -> m c) -> (a -> m b) -> (a -> m c)
f <=< g = (\x -> g x >>= f)
```


- the above is already defined in `Control.Monad`
- it helps establish associativity laws for Monads
  - the function `f` takes `b -> m c`
  - the function `g` takes `a -> m b`
- the problem is `g` outputs values of type the same as input for `f`, but monadic
- so `<=<` helps manage their composition

Associativity of `<=<`

Recall that the following are equivalent (associativity we should implement for `>>=`):

- `(m >>= f) >>= g`
- `m >>= (\x -> f x >>= g)`

Then it should also be that the following are equivalent:

- `(f <=< g) <=< h`

- `f <=< (g <=< h)`

Then we can also omit parentheses with chaining `<=<`.

But, you have to implement `>>=` properly!

— More Simplifications —

More Simplifications

Translating left identity laws:

- `` `return x >>= f` ``
- `` `f x` ``

For `<=<`, then we can also say the following:

- `` `f <=< return` `` is the same as just `` `f` ``

So, for right identity, `` `return <=< f` `` is also the same as `` `f` ``.

— Yo!, We Have Done a Lot —

Helpful Resources

For knowing exactly what thing you are working with and its corresponding documentation (like, which package?):

```
:info <name_of_thing>
```


A great way to style your error handling:

Gabriella Gonzalez (Google Blogger)

https://www.haskellforall.com/2021/05/the-trick-to-avoid-deeply-nested-error.html

Working with json style object initialization from files:

- grab the `json.zip` file from Blackboard
  - you will need to install the yaml package, but there are notes to help with the edits from the article

Names for Binary Operations

$	(none, just as " " [whitespace])	
->	to	a -> b: a to b
.	pipe to	a . b: "b pipe-to a"
<$>	(f)map	
<*>	ap(ply)	(as it is the same as Control.Monad.ap)
>>=	bind	
<-	bind	(as it desugars to >>=)
>>	then	
!!	index	
[]	empty list	
:	cons	
\	lambda	
@	as	go ll@(l:ls): go ll as l cons ls
::	of type / as	f x :: Int: f x of type Int

(others we have not covered)

*>	then (evaluates to right hand functor, unless left mempty)	
<$	map-replace by (e.g.:	0 <$ f: "f map-replace by 0“ 3 <$ [2] evaluates to [3])
<|>	or / alternative	expr <|> term: "expr or term“ (import Control.Applicative)
~	lazy (or irrefutable pattern) (https://en.wikibooks.org/wiki/Haskell/Laziness#Lazy_pattern_matching)	
!	strict	(use in signatures) (causes pattern matching errors even for _)

https://stackoverflow.com/questions/7746894/are-there-pronounceable-names-for-common-haskell-operators

Why Did We Learn Haskell?

https://crypto.stanford.edu/~blynn/haskell/why.html

Thank You!

Questions?